

ZE-AbqGetFil Ver.0.1

ABAQUS 板要素モデル fil ファイル読み込み例 Excel マクロ使用説明書

ソフトの概要

ABAQUS のテキスト形式 fil ファイルの読み込み例の Excel マクロ使用例の説明書です。例として作成した板要素モデルの応力計算結果を読み込み Excel シートに出力するマクロです。

ABAQUS の説明書に記載している Fortran での読み込みと同様にデータのキーワードを判断して処理します。

マクロを公開しますので自由に使用して下さい。なお、例として Excel シートに出力していますが Excel に出力できる最大行数はバージョンにより異なりますが非常に少ないので注意して下さい。

また、fil ファイルの読み込みは時間がかかりファイルが大きい計算結果には適していません。

大きいモデルのデータを処理する場合は、「Visual Studio」を使用し fil ファイルの読み込みルーチンを少し修正して使用して下さい。

出力例

- ①「要素」シート : 要素データ
- ②「節点」シート : 節点座標値データ
- ③「応力」シート : 板要素応力計算結果

注意事項

- ①本 VBA は個人所有ですが、参考にして改編を行って構いません。
- ②テストモデルでの機能のチェックは行っていますが、処理できないデータもあると思います。また、エラー処理等は十分に行っていません。
使用者の責任において使用して下さい。
- ③不具合の修正、機能の追加については使用者が行って下さい。
- ④本 VBA の内容についての質問は受け付けません。ただし、費用が発生しても問題なければ質問、機能の追加または作成については受け付けます。

開発者: 茨木 栄

Mail: sakae-ibaraki@mbr.nifty.com

改正来歴

説明書 Rev.	システム Ver.	改正日	改正内容
0.0	0.1	2021/11/26	初版

目次

1. 概要	1
2. 使用例	2
3. 計算結果	3
4. Excel VBA リスト	6

添付ファイル

①使用説明書

ZE-AbqGetFil_Ver. 0.1_ABAQUS 板要素モデル fil ファイル読み込み例
Excel マクロ使用説明書.docx

②VBA ファイル

ZE-AbqGetFil_Ver. 0.1.xlsm

③テストファイル

・ ABAQUS 関連ファイル

¥TestData¥testshell.fil

¥TestData¥TestShell.inp

¥TestData¥testshell.dat

1. 概要

ABAQUS の計算結果ファイルの fil ファイルはバイナリおよびテキスト形式のファイルがあります。Fortran での fil ファイルの読み込み方法は ABAQUS のマニュアルに記載しており ABAQUS のライブラリを使用して読み込みが可能です。しかし、この方法で作成された実行ファイルは ABAQUS が使用できる環境でないと実行できません。

アウトプットリストを読み込む方法もありますが、数値の桁数が少なくポスト処理に必要な有効桁数ではありません。バイナリファイルは出力形式が不明でありハードに依存するためテキスト形式のファイルを読み込むことにしました。

テキスト形式の fil ファイルは見た目は複雑であるが一定の法則で出力されているため Excel のマクロで読み込みシートに出力する本マクロを作成した。

本マクロでは読み込み例であり、板要素モデルの要素、節点及び応力をシートに出力する。ABAQUS のポストファイルはデータ内容によって処理が異なるため、処理の内容により作成する必要があります。

なお、fil ファイルの読み込みは ABAQUS のマニュアルに記載してある fil ファイルの読み込み方法同様にキーコードを判断してデータを読み込みます。

2. 使用例

本マクロは以下の手順で使用して下さい。

①ファイルのオープン

「ZE-AbqGetFil_Ver. 0.1.xlsm」を開くと以下のようにセルに何も無い状態でシートが表示されます。




図 2-1 Excel ファイルのオープン

②マクロの実行

図 2-1 で「Alt」+「F8」キーを押すと図 2-2 が表示されるので「実行」ボタンをクリックして下さい。




図 2-2 マクロの実行

③マクロの終了

マクロが終了すると図 2-3 が表示されるので「OK」ボタンをクリックして下さい。




図 2-3 マクロの終了

3. 計算結果

図 3-1～図 3-3 に fil ファイルを読み込んだ結果を Excel シートに出力した要素、節点および応力計算結果を示す。また、図 3-1～図 3-3 に対応する ABAQUS のインプットデータおよびアウトプットリストを表 3-1～表 3-3 に示す。

	1	2	3	4	5	6	7	8
1	4 S4		12	5	4	11		
2	5 S4		13	6	5	12		
3	6 S4		14	7	6	13		
4	10 S4		19	12	11	18		
5	11 S4		20	13	12	19		
6	12 S4		21	14	13	20		
7	16 S4		26	19	18	25		
8	17 S4		27	20	19	26		
9	18 S4		28	21	20	27		
10	22 S4		33	26	25	32		
11	23 S4		34	27	26	33		
12	24 S4		35	28	27	34		
13	28 S4		40	33	32	39		
14	29 S4		41	34	33	40		
15	30 S4		42	35	34	41		

図 3-1 「要素」シートの結果

表 3-1 ABAQUS インプットデータの要素データ

*ELEMENT, TYPE=S4, ELSET=P1				
4,	12,	5,	4,	11
5,	13,	6,	5,	12
6,	14,	7,	6,	13
10,	19,	12,	11,	18
11,	20,	13,	12,	19
12,	21,	14,	13,	20
16,	26,	19,	18,	25
17,	27,	20,	19,	26
18,	28,	21,	20,	27
22,	33,	26,	25,	32
23,	34,	27,	26,	33
24,	35,	28,	27,	34
28,	40,	33,	32,	39
29,	41,	34,	33,	40
30,	42,	35,	34,	41

	1	2	3	4	5	6	7	8
1	4	1472.296	193.8315	0				
2	5	1462.44	257.8677	0				
3	6	1452.549	308.7484	0				
4	7	1440.889	359.253	0				
5	11	1472.296	193.8315	101.6				
6	12	1462.44	257.8677	101.6				
7	13	1452.549	308.7484	101.6				
8	14	1440.889	359.253	101.6				
9	18	1472.296	193.8315	208.52				
10	19	1462.44	257.8677	208.52				
11	20	1452.549	308.7484	208.52				
12	21	1440.889	359.253	208.52				
13	25	1472.296	193.8315	315.48				
14	26	1462.44	257.8677	315.48				
15	27	1452.549	308.7484	315.48				

図 3-2 「節点」シートの結果

表 3-2 ABAQUS インプットデータの節点データ

*NODE,	NSET=NODEALL			
4,	1472.2956,	193.8315,	0.	
5,	1462.4395,	257.86769,	0.	
6,	1452.5493,	308.74841,	0.	
7,	1440.8894,	359.25297,	0.	
11,	1472.2956,	193.8315,	101.6	
12,	1462.4395,	257.86769,	101.6	
13,	1452.5493,	308.74841,	101.6	
14,	1440.8894,	359.25297,	101.6	
18,	1472.2956,	193.8315,	208.52	
19,	1462.4395,	257.86769,	208.52	
20,	1452.5493,	308.74841,	208.52	
21,	1440.8894,	359.25297,	208.52	
25,	1472.2956,	193.8315,	315.48	
26,	1462.4395,	257.86769,	315.48	
27,	1452.5493,	308.74841,	315.48	

	1	2	3	4	5	6	7	8
1	STEP		1 INCREMEN	1				
2	4	1	-0.16451	-0.04471	0.015081			
3	4	5	0.169336	0.044025	-0.01647			
4	5	1	-0.02137	-0.0239	0.047751			
5	5	5	0.018568	0.022611	-0.05064			
6	6	1	-0.00316	-0.02164	0.017995			
7	6	5	0.002165	0.024939	-0.01983			
8	10	1	-0.1777	-0.05606	-0.04402			
9	10	5	0.185195	0.057658	0.047898			
10	11	1	0.013658	-0.02007	-0.05338			
11	11	5	-0.01895	0.018509	0.06002			
12	12	1	0.002749	-0.01373	-0.02889			
13	12	5	-0.0041	0.013451	0.032797			
14	16	1	-0.02692	0.025779	-0.00404			
15	16	5	0.028283	-0.02243	0.003512			

図 3-3 「応力」シートの結果

表 3-3 ABAQUS アウトプットリストの応力計算結果

THE FOLLOWING TABLE IS PRINTED FOR ALL ELEMENTS WITH TYPE S4 AT THE CENTROID OF THE ELEMENT

ELEMENT	SEC	FOOT-PT	NOTE	S11	S22	S12
4	1			-0.1645	-4.4709E-02	1.5081E-02
4	5			0.1693	4.4025E-02	-1.6475E-02
5	1			-2.1371E-02	-2.3897E-02	4.7751E-02
5	5			1.8568E-02	2.2611E-02	-5.0638E-02
6	1			-3.1616E-03	-2.1644E-02	1.7995E-02
6	5			2.1653E-03	2.4939E-02	-1.9826E-02
10	1			-0.1777	-5.6063E-02	-4.4020E-02
10	5			0.1852	5.7658E-02	4.7898E-02
11	1			1.3658E-02	-2.0071E-02	-5.3382E-02
11	5			-1.8948E-02	1.8509E-02	6.0020E-02
12	1			2.7490E-03	-1.3730E-02	-2.8887E-02
12	5			-4.0961E-03	1.3451E-02	3.2797E-02
16	1			-2.6923E-02	2.5779E-02	-4.0414E-03
16	5			2.8283E-02	-2.2426E-02	3.5121E-03

4. Excel VBA リスト

表 4-1 に本 VBA のサブルーチン一覧表を、表 4-2～表 4-6 に本資料で使用した VBA のリストを示す。

本マクロは ABAQUS の fil ファイル名は固定としています。ファイル名を変更する場合は表 4-3 内の変数「FilFile」を使用する fil ファイル名に変更して下さい。また、本マクロでは fil ファイルのデータ内容を確認するためテキストファイルに出力ができます。

表 4-3 内の変数「ChkFile」がチェックリスト名となり、変数「iPutFile」を 0 と 1 以外の数値に変更して下さい。

なお、改変する場合は表 4-3 に示す「SampleGetFil」を参考に改変して下さい。表 4-4 に示す「GetFilFil」と表 4-5 に示す「GetFilData」は不具合がなければ改変する必要はありません。

表 4-1 サブルーチン一覧表

No.	サブルーチン名	内容
1		定数
2	SampleGetFil	要素、節点及び要素応力データの読み込み例
3	GetFilFile	fil ファイルデータ内のデータの読み込み
4	GetFilData	fil ファイルデータの読み込み（1 データ）

表 4-2 定数

```
' *****
' ABAQUS Fil ファイル(ASCII)読みマクロ
' 作成日 2021/8/29
' 作成者 茨木 栄
' *****
Option Explicit
Const MAX_ARRAY = 513
```

表 4-3 要素、節点及び要素応力データの読み例(1/2)

```

'要素、節点及び要素応力データの読み例
Sub SampleGetFil()
  Dim i As Long
  Dim Array(MAX_ARRAY) As Double '実数データ
  Dim jArray(2, MAX_ARRAY) As Long '整数データ
  Dim Word(MAX_ARRAY) As String '文字列データ
  Dim Key As Integer 'コード番号

  Dim FilFile As String 'ABAQUS Fil ファイル名
  Dim ChkFile As String 'チェックリスト名
  Dim iGetFile As Integer 'ABAQUS Fil ファイル ファイル番号
  Dim iPutFile As Integer 'チェックリスト ファイル番号 0にすると出力しない

  '要素変数の先頭のデータ
  Dim id As Long '要素番号(IDPOINT<>4), 節点番号(IDPOINT=4)
  Dim IDN As Integer '節点番号(IDPOINT=2)
  Dim IDP As Integer '断面番号
  Dim idpoint '出力点の識別番号 0=積分点, 1=要素中心, 2=節点, 4=節点平均

  Dim nRowElem As Long
  Dim nRowNode As Long
  Dim nRowStre As Long
  Dim ElemType As String * 8

  'ファイルの設定
  FilFile = ActiveWorkbook.Path & "%TestData%testshell.fil" '<=== File ファイル名
  ChkFile = ActiveWorkbook.Path & "%TestData%Check.lst" '<=== チェックリストファイル名

  iGetFile = 1
  iPutFile = 2 'チェックリストを出力 0:出力しない、0,1以外:出力する

  'ファイルのオープン
  Open FilFile For Input As #iGetFile
  If iGetFile <> iPutFile And iPutFile <> 0 Then
    Open ChkFile For Output As #iPutFile
  End If

  '-----
  ' ABAQUS Fil ファイルデータの読み
  '-----
  Do While Not EOF(1) ' ファイルの終端かどうかを確認します。

    ' Fil ファイルデータ内のデータの読み
    Call GetFilFile(iGetFile, iPutFile, Array, jArray, Word)

    Key = jArray(1, 2) ' Key によりデータを識別する

  '-----
  ' 要素データをシートに設定
  '-----
  If Key = 1900 Then '要素データ
    nRowElem = nRowElem + 1
    Sheets(1).Cells(nRowElem, 1).Value = jArray(1, 3)
    Sheets(1).Cells(nRowElem, 2).Value = Word(4)
    ElemType = Word(4)
    For i = 1 To 20
      If jArray(1, 4 + i) = 0 Then Exit For
      Sheets(1).Cells(nRowElem, 2 + i) = jArray(1, 4 + i)
    Next
  End If

```

表 4-4 要素、節点及び要素応力データの読み込み例(2/2)

```

' -----
' 節点データをシートに設定
' -----
If Key = 1901 Then ' 節点データ
  nRowNode = nRowNode + 1
  Sheets(2).Cells(nRowNode, 1).Value = jArray(1, 3)
  For i = 1 To 3
    Sheets(2).Cells(nRowNode, 1 + i) = Aray(3 + i)
  Next
End If

' -----
' ステップ、インクリメント番号をシートに設定
' -----
If Key = 2000 Then ' インクリメントの最初
  nRowStre = nRowStre + 1
  Sheets(3).Cells(nRowStre, 1).Value = "STEP"
  Sheets(3).Cells(nRowStre, 2).Value = jArray(1, 8)
  Sheets(3).Cells(nRowStre, 3).Value = "INCREMENT"
  Sheets(3).Cells(nRowStre, 4).Value = jArray(1, 9)
End If

' -----
' 要素変数の先頭データの取得
' -----
If Key = 1 Then ' 要素変数の先頭
  iD = jArray(1, 3) ' 要素番号 (IDPOINT<>4), 節点番号 (IDPOINT=4)
  IDN = jArray(1, 4) ' 節点番号 (IDPOINT=2)
  IDP = jArray(1, 5) ' 断面番号
  idpoint = jArray(1, 6) ' 出力点の識別番号 0=積分点, 1=要素中心, 2=節点, 4=節点平均
End If

' -----
' 要素応力をシートに設定 S4 タイプのみ
' -----
' 要素応力は要素タイプによって異なる
If Key = 11 Then ' 要素応力
  If ElemType = "S4" And idpoint = 1 Then
    nRowStre = nRowStre + 1
    Sheets(3).Cells(nRowStre, 1).Value = iD
    Sheets(3).Cells(nRowStre, 2).Value = IDP
    For i = 1 To 3
      Sheets(3).Cells(nRowStre, 2 + i) = Aray(2 + i)
    Next
  End If
End If
Loop

' -----
' 終了
' -----
' ファイルを閉じます。
Close #iGetFile
If iGetFile <> iPutFile And iPutFile <> 0 Then
  Close #iPutFile
End If

MsgBox "正常に終了しました。", vbOKOnly, "終了"
End Sub

```

表 4-5 Fil ファイルデータ内のデータの読み込み

```

'*****
'Fil ファイルデータ内のデータの読み込み
'*****
Sub GetFilFile(iGetFile, iPutFile, Array, jArray, Word)
  Dim Card As String
  Static Rec As String

  Do
    Line Input #iGetFile, Card 'データ行を読み込みます。
    Rec = Rec & Card '読込んだデータを前のデータの後ろに付ける
    If EOF(iGetFile) Or InStr(2, Rec, "*I") <> 0 Then Exit Do ' *I の最後まで読む
  Loop

  If EOF(iGetFile) Then '最後のデータ
    If iPutFile <> 0 Then
      Print #iPutFile, Rec
    End If

    '文字列からデータの読み込み
    Call GetFilData(Rec, Array, jArray, Word)
  Else
    If iPutFile <> 0 Then
      Print #iPutFile, Mid(Rec, 1, InStr(2, Rec, "*I") - 1)
    End If

    '文字列からデータの読み込み
    Call GetFilData(Mid(Rec, 1, InStr(2, Rec, "*I") - 1), Array, jArray, Word)

    '*I が最初になるように文字をシフト
    Rec = Mid(Rec, InStr(2, Rec, "*I"), Len(Rec) - InStr(2, Rec, "*I") + 1)
  End If
End Sub

```

表 4-6 Fil ファイルデータの読み込み (1 データ)

```

'*****
' Fil ファイルデータの読み込み (1 データ)
' ASCII コードの Fil ファイル例 A の後が文字、I の後が整数、E,D の後が実数
'-----
'*I 19I 41921A6.12-1 A12-9-201A6      A16:41:38I 3216I 3252D 8.476653841821714D
'+01*I 18I 41900I 14AS4      I 212I 15I 14I 211*I 18I 41900I 15AS4      I 213I 16
'408*I 18I 41900I 3402AS4      I 3249I 3242I 3406I 3409*I 19I 41901I 14D 1.472295
'600000000D+03D 1.938315000000000D+02D 0.000000000000000D+00D-9.914448847445233D-
'*****
Sub GetFilData(Rec, Array, jArray, Word)
  Dim i As Long
  Dim nc, ns As Long '文字位置

  'データのクリア
  For i = 0 To MAX_ARRAY
    Array(i) = 0# '実数
    jArray(2, i) = 0 '整数
    Word(i) = "" '文字列
  Next

  nc = Val(Mid(Rec, 3, 2)) 'データ数の文字数の最初の文字位置
  jArray(1, 1) = Val(Mid(Rec, 5, nc)) 'データ数の読み込み
  ns = 5 + nc '最初のデータの文字一の始まり位置

  'データの読み込み
  For i = 2 To jArray(1, 1) 'データ数
    Select Case Mid(Rec, ns, 1) 'データタイプ
      Case "A" '文字データ 文字は8文字で固定
        Word(i) = Mid(Rec, ns + 1, 8)
        ns = ns + 8 '最後の文字位置

      Case "I" '整数データ 整数は文字数を読み込みその字数分を読み込む
        nc = Val(Mid(Rec, ns + 1, 2))
        jArray(1, i) = Val(Mid(Rec, ns + 3, nc))
        ns = ns + 2 + nc '最後の文字位置

      Case "E", "D" '実数データ 実数は22文字で固定
        Array(i) = Val(Mid(Rec, ns + 1, 22))
        ns = ns + 22 '最後の文字位置
    End Select
    ns = ns + 1 '次のデータの最初の文字位置
  Next
End Sub

```